Universidad del País Vasco
Euskal Herriko Unibertsitatea

Ejercicios resueltos

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Tema 5. Teoría de la aproximación

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

EJERCICIOS DEL TEMA 5. TEORÍA DE LA APROXIMACIÓN

Ejercicio nº1

Enunciado

Determine si los siguientes subconjuntos del espacio vectorial euclídeo $\mathbb{R}^3$, con el producto escalar usual, son ortogonales y/o ortonormales.

- a) $S = \left\{\left(\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right), \left(\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right), (0, 1, 0)\right\}$
- b) $T = \left\{\left(0, \cos\left(\frac{\pi}{2}\right), 1\right), (-\cos(\pi), 0, 0), \left(0, \operatorname{sen}\left(\frac{\pi}{2}\right), 0\right)\right\}$
- c) $U = \{(2, 1, 0), (1, -2, 0), (0, 0, 3)\}$

Resolución

- a) $S = \left\{\left(\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right), \left(\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right), (0, 1, 0)\right\}$
  - Introducción del subconjunto S como una matriz:

```
S = {u1 = {1 / Sqrt[3], 1 / Sqrt[3], 1 / Sqrt[3]}, u2 = {1 / Sqrt[2], 0, -1 / Sqrt[2]}, u3 = {0, 1, 0}}
```

$$\left\{\left\{\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right\}, \left\{\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right\}, \{0, 1, 0\}\right\}$$

```
S // MatrixForm
```

$$\begin{pmatrix} \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{3}} \\ \frac{1}{\sqrt{2}} & 0 & -\frac{1}{\sqrt{2}} \\ 0 & 1 & 0 \end{pmatrix}$$

  - Obtención de la matriz de Gram:

```
ga = Table[S[[i]].S[[j]], {i, 1, Length[S]}, {j, 1, Length[S]}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & \frac{1}{\sqrt{3}} \\ 0 & 1 & 0 \\ \frac{1}{\sqrt{3}} & 0 & 1 \end{pmatrix}$$

  - La matriz de Gram no es diagonal: el sistema S no es ortogonal y, por tanto, tampoco ortonormal

- b) $T = \left\{\left(0, \cos\left(\frac{\pi}{2}\right), 1\right), (-\cos(\pi), 0, 0), \left(0, \operatorname{sen}\left(\frac{\pi}{2}\right), 0\right)\right\}$
  - Introducción del subconjunto T como una matriz:

```
T = {{0, Cos[Pi / 2], 1}, {-Cos[Pi], 0, 0}, {0, Sin[Pi / 2], 0}}
T // MatrixForm
```

$$\{\{0, 0, 1\}, \{1, 0, 0\}, \{0, 1, 0\}\}$$

$$\begin{pmatrix} 0 & 0 & 1 \\ 1 & 0 & 0 \\ 0 & 1 & 0 \end{pmatrix}$$

- Obtención de la matriz de Gram:

```
Table[T[[i]].T[[j]], {i, 1, Length[T]}, {j, 1, Length[T]}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

- La matriz de Gram es la matriz identidad: el sistema T es ortonormal y, por tanto, ortogonal

- c) $U = \{(2, 1, 0), (1, -2, 0), (0, 0, 3)\}$
  - Introducción del subconjunto U como una matriz:

```
U = {{2, 1, 0}, {1, -2, 0}, {0, 0, 3}};
```

```
U // MatrixForm
```

$$\begin{pmatrix} 2 & 1 & 0 \\ 1 & -2 & 0 \\ 0 & 0 & 3 \end{pmatrix}$$

- Obtención de la matriz de Gram:

```
Table[U[[i]].U[[j]], {i, 1, Length[U]}, {j, 1, Length[U]}] // MatrixForm
```

$$\begin{pmatrix} 5 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 9 \end{pmatrix}$$

- La matriz de Gram es diagonal, por tanto, el sistema de vectores U :
  - es ortogonal
  - no es ortonormal ya que la matriz no coincide con la identidad

Ejercicio nº2

Enunciado

En el espacio vectorial euclídeo $\left(\mathbb{R}^3, \ <\,,\ >\right)$ se considera el producto escalar:

$$<\vec{x}, \vec{y}> = x_1\, y_1 + x_3\, y_1 + 3\, x_2\, y_2 + x_1\, y_3 + 5\, x_3\, y_3 \text{ siendo } \vec{x} = (x_1, x_2, x_3)\,, \vec{y} = (y_1, y_2, y_3)$$

- a) Determine la norma del vector $\vec{v} = (-1, 1, 1)$.
- b) Calcule la norma del vector $\vec{v}$ utilizando el producto escalar usual.
- c) Compare los resultados obtenidos en los dos apartados anteriores.
- d) Obtenga la distancia entre los vectores $\vec{u} = (2, 0, -3)$ y $\vec{v} = (-1, 1, 1)$.
- e) Calcule el ángulo que forman ambos vectores.

Resolución

- a) Determine la norma del vector $\vec{v} = (-1, 1, 1)$.
  - Definición del producto escalar definido en el enunciado

```
pe[u_, v_] := u[[1]] * v[[1]] + u[[3]] * v[[1]] + 3 * u[[2]] * v[[2]] + u[[1]] * v[[3]] + 5 * u[[3]] * v[[3]]
```

- Introducción del vector $\vec{v}$ y el producto escalar definidos en el enunciado

```
v = {-1, 1, 1};
```

- Cálculo de la norma:

```
normv = Sqrt[pe[v, v]]
```

$\sqrt{7}$

- Se tiene: $\boxed{\|\vec{v}\| = \sqrt{7}}$

- b) Calcule la norma del vector $\vec{v}$ utilizando el producto escalar usual.
  - Cálculo de la norma con el producto escalar usual de $\mathbb{R}^3$:

```
normv = Norm[v]
```

$\sqrt{3}$

- En este caso: $\boxed{\|\vec{v}\| = \sqrt{3}}$

- c) Compare los resultados obtenidos en los dos apartados anteriores.
  - Como los productos escalares son diferentes, las normas y las distancias inducidas por dichos productos no coinciden, es decir, los resultados difieren entre sí
- d) Obtenga la distancia entre los vectores $\vec{u} = (2, 0, -3)$ y $\vec{v} = (-1, 1, 5)$.
  - Definición del vector $\vec{u}$:

```
u = {2, 0, -3};
```

- La distancia entre ambos vectores es la norma del vector $\vec{u} - \vec{v}$:

```
dist = Sqrt[pe[u - v, u - v]]
```

$2\sqrt{17}$

- Solución: $\boxed{d(\vec{u}, \vec{v}) = 2\sqrt{17}}$

- e) Calcule el ángulo que forman ambos vectores.
  - Cálculo de las normas de ambos vectores:

```
normu = Sqrt[pe[u, u]];  normv = Sqrt[pe[v, v]];
```

- Obtención del ángulo que forman los vectores $\vec{u}$ y $\vec{v}$:

```
ang = ArcCos[pe[u, v] / (normu * normv)] // N
```

2.4123

- El ángulo viene dado en radianes; para obtener el correspondiente valor en grados sexagesimales:

```
ang / Degree
```

138.214

- Solución: $\boxed{\begin{array}{c} ang(\vec{u}, \vec{v}) \approx 2.4123\, rad \\ ang(\vec{u}, \vec{v}) \approx 138.214^{o} \end{array}}$

Ejercicio nº3

Enunciado

Sea espacio vectorial euclídeo ($\mathbb{P}_2(x)$, $<\ ,\ >$) con el producto escalar:

$$< p(x), q(x) > \; = p(-1)\, q(-1) + p(0)\, q(0) + p(1)\, q(1)$$

- a) Calcule el producto escalar entre $p(x) = 1 + x^2$ y $q(x) = x - 2$.
- b) Considere la base canónica de $\mathbb{P}_2(x)$ y obtenga la matriz de Gram.
- c) Calcule el producto escalar $<p(x),q(x)>$ mediante la matriz de Gram.

Resolución

- a) Calcule el producto escalar entre $p(x) = 1 + x^2$ y $q(x) = x - 2$.
  - Definición del producto escalar definido en el enunciado

```
pe[p_, q_] := p[-1] * q[-1] + p[0] * q[0] + p[1] * q[1];
```

  - Introducción de los polinomios

```
p[x_] = 1 + x^2; q[x_] = x - 2;
```

  - Cálculo del producto escalar pedido

```
pe[p, q]
```

```
-10
```

  - Solución: $\boxed{< p(x), q(x) > \; = -10}$

- b) Considere la base canónica de $\mathbb{P}_2(x)$ y obtenga la matriz de Gram.
  - Introducción de la base canónica de $\mathbb{P}_2(x)$ como una matriz

```
B = {1, x, x^2};
```

  - Cálculo de la matriz de Gram

```
Gram = Table[p1[x_] = B[[i]];
  p2[x_] = B[[j]];
  pe[p1, p2], {i, 1, Length[B]}, {j, 1, Length[B]}]; Gram // MatrixForm
```

$$\begin{pmatrix} 3 & 0 & 2 \\ 0 & 2 & 0 \\ 2 & 0 & 2 \end{pmatrix}$$

  - Solución: $\boxed{G_B = \begin{pmatrix} 3 & 0 & 2 \\ 0 & 2 & 0 \\ 2 & 0 & 2 \end{pmatrix}}$

- c) Calcule el producto escalar $<p(x),q(x)>$ mediante la matriz de Gram.
  - Coordenadas de los polinomios en la base dada

```
{coordp, coordq} = CoefficientList[{p[x], q[x]}, x, 3];
```

  - Cálculo del producto escalar pedido:

```
coordp.Gram.coordq
```

```
-10
```

Ejercicio nº4

Enunciado

En el espacio vectorial euclídeo $(\mathbb{M}_{2\times 2}(\mathbb{R}), <,>)$ con el producto escalar usual se considera el subespacio vectorial $S = \{A \in \mathbb{M}_{2\times 2}(\mathbb{R}) / A^{T} = A\}$

- a) Calcule una base del subespacio S.
- b) Determine una base ortogonal del subespacio S.
- c) Obtenga una base ortonormal del subespacio S.
- d) Calcule la mejor aproximación de la matriz $C = \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix}$ en el subespacio S y estime el error cometido.

Resolución

- a) Calcule una base del subespacio S.
  - Definición de una matriz genérica del espacio $\mathbb{M}_{2\times 2}(\mathbb{R})$

```
A = {{a, b}, {c, d}};
```

  - Ecuación implícita del subespacio S

```
sol = Solve[Transpose[A] == A]
```

```
{{c → b}}
```

  - Forma general de una matriz perteneciente al subespacio S

```
A = A /. sol[[1]]; A // MatrixForm
```

$\begin{pmatrix} a & b \\ b & d \end{pmatrix}$

  - Obtención de una base de S dando valores a los parámetros a, b y d

```
b1 = A /. {a -> 1, b → 0, d -> 0}; b2 = A /. {a → 0, b → 1, d -> 0}; b3 = A /. {a → 0, b → 0, d → 1};
```

```
{b1 // MatrixForm, b2 // MatrixForm, b3 // MatrixForm}
```

$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}\right\}$

  - Solución: $\boxed{B = \left\{B_1 = \begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}, B_2 = \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, B_3 = \begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}\right\}}$

- b) Determine una base ortogonal del subespacio S.
  - Definición del producto escalar usual del espacio vectorial

```
pe[A_, B_] := Tr[Transpose[B].A]
```

  - Definición de la base obtenida en el apartado anterior y análisis de si es ortogonal

```
B = {b1, b2, b3};
Gram = Table[pe[B[[i]], B[[j]]], {i, 1, Length[B]}, {j, 1, Length[B]}];
Gram // MatrixForm
```

$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

  - La base es ortogonal ya que la matriz de Gram es diagonal

- c) Obtenga una base ortonormal del subespacio S.
  - Como la matriz de Gram no coincide con la identidad, la base es ortogonal pero no ortonormal
  - Se deduce de la matriz de Gram que el único vector no unitario de la base es $b2$
  - Por tanto, para ortonormalizar dicha base sólo se debe normalizar la matriz $b2$

```
b2p = b2 / Sqrt[pe[b2, b2]]; b2p // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & 0 \end{pmatrix}$$

  - La base ortonormal obtenida es: $\boxed{B_{ON} = \left\{B_1 = \begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}, B'_2 = \begin{pmatrix} 0 & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & 0 \end{pmatrix}, B_3 = \begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}\right\}}$
  - También, puede usarse la función **Orthogonalize**

```
{t1, t2, t3} = Orthogonalize[B, pe];
{t1 // MatrixForm, t2 // MatrixForm, t3 // MatrixForm}
```

$$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}\right\}$$

- d) Calcule la mejor aproximación de la matriz $C = \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix}$ en el subespacio S y estime el error cometido.
  - Matriz que se aproxima:

```
M = {{0, 1}, {-1, 0}};
```

  - Base ortogonal de S:

```
B = {b1, b2, b3};
```

  - La mejor aproximación de M en S es su proyección ortogonal en dicho subespacio:

```
aprox = Sum[Projection[M, B[[i]], pe], {i, Length[B]}]
```

```
{{0, 0}, {0, 0}}
```

  - Por tanto, la mejor aproximación es: $\boxed{proy_S\, M = \left\{ [0]_{2\times 2} = \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}\right\}}$
  - Este resultado se debe a que la matriz dada es ortogonal al subespacio: $M \perp S$

Ejercicio nº5

Enunciado

En el espacio vectorial euclídeo $\mathbb{R}^3$ con el producto escalar usual se considera el subespacio vectorial:

$$S = \left\{(x, y, z) \in \mathbb{R}^3 \,/\, z = 0\right\}$$

- a) Obtenga el subespacio $S^\perp$ ortogonal a S.
- b) Represente gráficamente S y $S^\perp$ y deduzca el subespacio $S \cap S^\perp$.
- c) Obtenga el subespacio vectorial $S + S^\perp$.

Resolución

- a) Obtenga el subespacio $S^{\perp}$ ortogonal a S.
  - El subespacio ortogonal a S está formado por todos aquellos vectores ortogonales a S y, por tanto, ortogonales a una base cualquiera de S
  - Se proporciona la ecuación implícita de S (se trata del plano $z = 0$)
  - $B = \{(1, 0, 0), (0, 1, 0)\}$ es una base de S

```
b1 = {1, 0, 0}; b2 = {0, 1, 0}; B = {b1, b2}; MatrixRank[B]
```

```
2
```

  - Se exige que un vector genérico $\vec{v}(x, y, z) \in \mathbb{R}^3$ sea ortogonal a la base B :

```
v = {x, y, z};
Solve[Table[Dot[v, B[[i]]] == 0, {i, Length[B]}]]
```

```
{{x → 0, y → 0}}
```

  - Ecuaciones implícitas de S (eje OZ): $\boxed{S^{\perp} = \{(x, y, z) / x = y = 0\}}$

- b) Represente gráficamente S y $S^{\perp}$ y deduzca el subespacio $S \cap S^{\perp}$.

```
Graphics3D[{{Black, Thick, Line[{{0, 0, -1}, {0, 0, 1}}]}, {Red, PointSize[Large], Point[{0, 0, 0}]},
  {Green, Opacity[.5], Polygon[{{2, 2, 0}, {-2, 2, 0}, {-2, -2, 0}, {2, -2, 0}}]}}, Axes → True,
 AxesOrigin → Automatic, AxesLabel → {x, y, z}, Boxed → False, Ticks → {{0, 2}, {0, 2}, {0, 2}}]
```



  - Evidentemente: $\boxed{S \cap S^{\perp} = \{(0, 0, 0)\}}$

- c) Obtenga el subespacio vectorial $S+S^{\perp}$.
  - De los resultados anteriores y la representación gráfica se deduce que: $\boxed{S + S^{\perp} = \mathbb{R}^3}$
  - Es más: $\boxed{S \oplus S^{\perp} = \mathbb{R}^3}$

Ejercicio nº6

Enunciado

En el espacio vectorial euclídeo $\mathbb{R}^4$ con el producto escalar usual se considera el siguiente sistema de ecuaciones lineales:

$$\begin{cases} x_1 + 2\,x_2 - x_3 = 1 \\ -x_1 - x_2 - x_3 = 0 \\ x_1 + x_3 = 1 \\ x_1 - x_3 = 0 \end{cases}$$

- a) Resuelva el sistema si es posible.
- b) Resuelva el sistema de forma aproximada (en el sentido de mínimos cuadrados).

Resolución

- a) Resuelva el sistema si es posible.
  - Expresión vectorial del sistema: $\vec{a}_1$, $\vec{a}_2$, $\vec{a}_3$, $\vec{X}$ y $\vec{b}$

```
{{a1 = {1, -1, 1, 0}, a2 = {2, -1, 0, 0}, a3 = {-1, -1, 1, -1}}, X = {x1, x2, x3}, b = {1, 0, 1, 0}};
```

  - Sistema incompatible

```
A = Transpose[{a1, a2, a3}];

MatrixForm[A]
```

$$\begin{pmatrix} 1 & 2 & -1 \\ -1 & -1 & -1 \\ 1 & 0 & 1 \\ 0 & 0 & -1 \end{pmatrix}$$

```
Solve[A.X == b, X]

{}
```

  - *Mathematica* no devuelve resultado alguno, por lo que el sistema es incompatible
- b) Resuelva el sistema de forma aproximada (en el sentido de mínimos cuadrados).
  - El sistema es incompatible; por lo tanto, el rango de la matriz de coeficientes no coincide con el rango de la matriz ampliada o, lo que es lo mismo, el término independiente del sistema ($\vec{b}$) no es combinación lineal de los vectores columna de la matriz de coeficientes ($\vec{a}_1$,$\vec{a}_2$,$\vec{a}_3$)
  - Es decir: $\vec{b} \notin \mathcal{L}\left(\{\vec{a}_1, \vec{a}_2, \vec{a}_3\}\right) = W$
  - Para resolver el sistema de forma aproximada en el sentido de mínimos cuadrados ordinarios se debe reemplazar el término independiente $\vec{b}$ por su mejor aproximación en el subespacio W

$$S' \equiv \; x_1 \cdot \vec{a}_1 + x_2 \cdot \vec{a}_2 + x_3 \cdot \vec{a}_3 = \vec{b}' = proy_W \vec{b} \in W$$

  - Se estudia si la familia generadora es libre:

```
Solve[α * a1 + β * a2 + γ * a3 == {0, 0, 0, 0}] (* familia libre: el sistema es compatible determinado *)

{{α → 0, β → 0, γ → 0}}
```

  - Base ortogonal del subespacio W :

```
B = {a1, a2, a3};
bo = Orthogonalize[B]
```

$$\left\{\left\{\frac{1}{\sqrt{3}}, -\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, 0\right\}, \left\{\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}, 0\right\}, \left\{-\frac{1}{\sqrt{15}}, -\frac{2}{\sqrt{15}}, -\frac{1}{\sqrt{15}}, -\sqrt{\frac{3}{5}}\right\}\right\}$$

  - Mejor aproximación del vector $\vec{b}$ en el subespacio $\mathcal{L}\left(\{\vec{a}_1, \vec{a}_2, \vec{a}_3\}\right) = W$

```
d = Sum[Projection[b, bo[[i]]], {i, Length[bo]}] // Simplify
```

$$\left\{\frac{4}{5}, -\frac{2}{5}, \frac{4}{5}, \frac{2}{5}\right\}$$

  - Se resuelve el sistema S':

```
Solve[a1 * x1 + a2 * x2 + a3 * x3 == d]
```

$\left\{\left\{x1 \to \frac{6}{5},\ x2 \to -\frac{2}{5},\ x3 \to -\frac{2}{5}\right\}\right\}$

- Solución aproximada en el sentido de mínimos cuadrados: $\boxed{\begin{cases} x_1 = \frac{6}{5} \\ x_2 = -\frac{2}{5} \\ x_3 = -\frac{2}{5} \end{cases}}$

Ejercicio nº7

Enunciado

Se considera el espacio vectorial euclídeo, (C[0,1] , •), de las funciones continuas en el intervalo [0,1] con el producto escalar usual:

$$\forall\ f(x),\ g(x) \in \mathrm{C}[0,1]:\ f(x) \bullet g(x) = \int_0^1 f(x) \cdot g(x)\, dx$$

Aproxime, estimando el error cometido, la función $f(x) = e^x$ en el intervalo [0,1] mediante un polinomio:

- a) de grado dos.
- b) de grado tres.

Resolución

- a) Aproximación con un polinomio de grado dos.
  - Base usual del subespacio vectorial $S = \mathbb{P}_2(x) \subset C[0,1]$:

```
base = {f1[x_] = 1, f2[x_] = x, f3[x_] = x^2};
```

  - Definición del producto escalar usual del espacio vectorial euclídeo $C[0,1]$:

```
pe[f_, g_] := Integrate[f * g, {x, 0, 1}]
```

  - Base ortonormal del subespacio vectorial $S = \mathbb{P}_2(x) \subset C[0,1]$:

```
bort = Orthogonalize[base, pe] // Simplify
```

$\left\{1,\ \sqrt{3}\,(-1 + 2x),\ \sqrt{5}\,(1 - 6x + 6x^2)\right\}$

  - Mejor aproximación de $f(x) = e^x$ en S :

```
f[x_] := E^x;
```

```
vp[x_] = Sum[Projection[f[x], bort[[i]], pe], {i, Length[bort]}] // Simplify
```

$3\,(-35 + 196x - 190x^2 + e\,(13 - 72x + 70x^2))$

```
N[vp[x]] // Expand
```

$1.01299 + 0.851125\,x + 0.839184\,x^2$

  - Polinomio calculado para la aproximación:

$$\boxed{p(x) = 1.01299 + 0.851125\,x + 0.839184\,x^2}$$

- Representación gráfica y norma del error cometido:

```
e[x_] := f[x] - vp[x]
```

```
graferror = Plot[e[x], {x, 0, 1}, PlotStyle -> {Green, Thickness[0.01]}]
```




```
ne = Sqrt[pe[e[x], e[x]]] // N
```

```
0.00527593
```

$$\boxed{||error|| = 0.00527593}$$

- Representación gráfica de la función y el polinomio:
  - en el intervalo [0,1]

```
grafv = Plot[f[x], {x, 0, 1}, PlotStyle → {Red, Thickness[0.01]}];
```

```
grafvp = Plot[vp[x], {x, 0, 1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];
```

```
Show[grafv, grafvp]
```



  - en el intervalo [0,0.1]

```
gv = Plot[f[x], {x, 0, 0.1}, PlotStyle → {Red, Thickness[0.01]}];
```

```
gvp = Plot[vp[x], {x, 0, 0.1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];
```

```
Show[gv, gvp]
```



  - en el intervalo [0.9,1]

```
gv2 = Plot[f[x], {x, 0.9, 1}, PlotStyle → {Red, Thickness[0.01]}];
```

```
gvp2 = Plot[vp[x], {x, 0.9, 1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];
```

```
Show[gv2, gvp2]
```



- b) Aproximación con un polinomio de grado tres.
  - Base usual del subespacio vectorial $T = \mathbb{P}_3(x) \subset C[0, 1]$:

```
base = {f1[x_] = 1, f2[x_] = x, f3[x_] = x^2, f4[x_] = x^3}
```

$\{1, x, x^2, x^3\}$

  - Definición del producto escalar usual del espacio vectorial euclídeo $C[0,1]$:

```
pe[f_, g_] := Integrate[f*g, {x, 0, 1}]
```

  - Base ortonormal del subespacio vectorial $T = \mathbb{P}_3(x) \subset C[0, 1]$:

```
bort = Orthogonalize[base, pe] // Simplify
```

$\left\{1, \sqrt{3}\,(-1 + 2x), \sqrt{5}\,(1 - 6x + 6x^2), \sqrt{7}\,(-1 + 12x - 30x^2 + 20x^3)\right\}$

  - Mejor aproximación de $f(x) = e^x$ en T:

```
f[x_] := E^x;
```

```
vp[x_] = Sum[Projection[f[x], bort[[i]], pe], {i, Length[bort]}] // Simplify
```

$-4\left(364 - 4200x + 10275x^2 - 6755x^3 + e\left(-134 + 1545x - 3780x^2 + 2485x^3\right)\right)$

```
N[vp[x]] // Expand
```

$0.99906 + 1.0183x + 0.421246x^2 + 0.278625x^3$

  - Polinomio calculado para la aproximación:

$$\boxed{q(x) = 0.99906 + 1.0183\,x + 0.421246\,x^2 + 0.278625\,x^3}$$

  - Representación gráfica y norma del error cometido:

```
e[x_] := f[x] - vp[x]
```

```
graferror = Plot[e[x], {x, 0, 1}, PlotStyle -> {Green, Thickness[0.01]}]
```




```
ne = Sqrt[pe[e[x], e[x]]] // N
```

0.000331242

$$\boxed{||error|| = 0.000331242}$$

- Representación gráfica de la función y el polinomio:
  - en el intervalo [0,1]

```
grafv = Plot[f[x], {x, 0, 1}, PlotStyle → {Red, Thickness[0.01]}];

grafvp = Plot[vp[x], {x, 0, 1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];

Show[grafv, grafvp]
```



  - en el intervalo [0,0.1]

```
gv = Plot[f[x], {x, 0, 0.1}, PlotStyle → {Red, Thickness[0.01]}];

gvp = Plot[vp[x], {x, 0, 0.1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];

Show[gv, gvp]
```



  - en el intervalo [0.9,1]

```
gv2 = Plot[f[x], {x, 0.9, 1}, PlotStyle → {Red, Thickness[0.01]}];

gvp2 = Plot[vp[x], {x, 0.9, 1}, PlotStyle → {Blue, Dashing[0.05], Thickness[0.015]}];

Show[gv2, gvp2]
```